

MASSACHUSETTS INSTITUTE OF TECHNOLOGY
HAYSTACK OBSERVATORY
WESTFORD, MASSACHUSETTS 01886

June 26, 2006

Telephone: 978-692-4764
Fax: 781-981-0590

To: SRT Group

From: Alan E.E. Rogers

Subject: Firmware and software changes to support “fractional” counts from the motor drives

The SRT position encoding originally assumed the following:

- 1] The drive reference position when turning on the motor results in no pulses from the reed switch within the stamp PULSIN timeout of 655 ms
- 2] When a move of N counts are required and the motor is turned on until N counts are received the motor will have moved the magnetic wheel through N sectors of the magnetic disk. In other words it is assumed that the motor will not coast into the next sector.

The second assumption resulted in very simple code and the experience with the H180s, as well as tests made looking at how far the magnetic wheel moved after removing power and shorting the terminal, proved the simple approach to be adequate.

The new mount from Cassicorp uses a superjack screwjack (or pushrod) in elevation. There are no specifications on how much the screw will “coast” and initial tests did not indicate that there was a problem although no measurements were made on how far the drive would coast.

In order to keep track on the pulses, on the assumption that there may be considerable coasting a more sophisticated approach has been adopted as follows:

- 1] The PULSIN routine counts pulses that start with a zero to one transition. If a pulse start and a pulse end (one to zero) are not seen within the timeout period of 655 ms a pulse is not counted.
- 2] When the desired number of pulses has been reached the motor is shut down.
- 3] The PULSIN routine continues but will timeout unless additional pulses are detected during the time the motor continues to rotate to an eventual stop.
- 4] Under normal conditions the reed switch will be closed in the low state at the start of the move operating and be in the low state at the end of the move. If the motor coasts to a stop and leaves the reed switch open we count this as a “half” pulse so the algorithm is completed by returning 2 numbers to the high level software.
 - a. The number of counts accumulated in the PULSIN loop.
 - b. The number of “half” counts during the move operation. If the read switch is open (high state) at the start of the move this counts as one half pulse and if the switch is open at the end of the move another half is counted. i.e. a number of zero, 1 or 2 is returned.
- 5] The high level software keep track of the number “half” counts equal to twice the number of counts plus the number of half counts. A listing of the revised stamps code follows:

```

' {$STAMP BS1}
DIRS = %00000000
LOW 7          'az motor
LOW 6          'H-bridge polarity
LOW 5          'el motor
LOW 2          ' cal power

Main:
W2 = 0
B2 = 0
SERIN 0,N2400,("move"),#B8,#W5
'b8=2
'w5=100
'b8=1
'debug b8,w5
BRANCH B8, (M1CW,M1CCW,M2CW,M2CCW,CALOUT,CALIN,NCALOFF,NCALON)
SEROUT 1,N2400,("E ",#B8," ",#W5,13)
DIR1 = 0          'open output
GOTO Main

Pulse1:

IF PIN3 = 0 THEN Ps1
B2 = 1          'starts in next sw zone
Ps1:
HIGH 7          'turn on motor
Pls1:
PULSIN 3,1,W0
IF W0=0 THEN Tm1
W2=W2+1
IF W2<W5 THEN Pls1
LOW 7          'turn off motor
GOTO Pls1
Tm1:
IF W2<W5 THEN Timeout
IF PIN3 = 0 THEN MOUT
B2=B2+1          'passed to next sw zone add one
GOTO MOUT

Pulse2:

IF PIN4 = 0 THEN Ps2
B2 = 1
Ps2:
HIGH 5
Pls2:
PULSIN 4,1,W0
IF W0=0 THEN Tm2
W2=W2+1
IF W2<W5 THEN Pls2
LOW 5
GOTO Pls2
Tm2:
IF W2<W5 THEN Timeout
IF PIN4 = 0 THEN MOUT
B2=B2+1

```

MOUT:

```
SEROUT 1,N2400,("M ",#W2," ",#B2," ",#B8,13)
DIR1 = 0
GOTO Main
```

M2CW:

```
LOW 6
GOTO Pulse2
```

M2CCW:

```
HIGH 6
GOTO Pulse2
```

M1CW:

```
LOW 6
GOTO Pulse1
```

M1CCW:

```
HIGH 6
GOTO Pulse1
```

Timeout:

```
PAUSE 50
SEROUT 1,N2400,("T ",#W2," ",#B8," ",#W5,13)
DIR1 = 0
W5=0
LOW 7
LOW 6
LOW 5
GOTO Main
```

NCALOFF:

```
LOW 2
GOTO Sout
```

NCALON:

```
HIGH 6
HIGH 2
GOTO Sout
```

CALIN:

```
LOW 6
HIGH 2
PAUSE 20000
LOW 2
GOTO Sout
```

CALOUT:

```
HIGH 6
HIGH 2
PAUSE 20000
```

```
LOW 2  
GOTO Sout
```

```
Sout:
```

```
SEROUT 1,N2400,("C ",#W2," ",#B8," ",#W5,13)  
DIR1=0  
GOTO Main
```